多段階選抜 (Pike)


|多段階選抜|解答日|シリーズ:yieldの練習/ジェネレータを入れ子に/整数平方根・立方根の実装|
|:--|:--|:--|
|問題|　|http://nabetani.sakura.ne.jp/hena/ord24eliseq/<br>https://qiita.com/Nabetani/items/1c83005a854d2c6cbb69|
|Ruby|2014/8/2(当日)|https://qiita.com/cielavenir/items/9f15e29b73ecf98968a5|
|C#/Python|2014/8/4|https://qiita.com/cielavenir/items/a1156e6a4f71ddbe5dcb|
| |　|ここから上はdrop_prev_square/drop_prev_cubicをまとめる前の答案|
|Go/C#/Ruby/Python|2014/8/5|https://qiita.com/cielavenir/items/2a685d3080862f2c2c47|
|PHP/JavaScript|2014/9/9|https://qiita.com/cielavenir/items/28d613ac3823afbf8407|
|VB|2014/9/10|https://qiita.com/cielavenir/items/cb7266abd30eadd71c04|
|D|2015/12/21|https://qiita.com/cielavenir/items/47c9e50ee60bef2847ec|
|Perl|2017/3/10|https://qiita.com/cielavenir/items/6dfbff749d833c0fd423|
|Lua|2017/3/13|https://qiita.com/cielavenir/items/c60fe7e8da73487ba062|
|C++20(TS)|2017/3/15|https://qiita.com/cielavenir/items/e1129ca185008f49cbab (MSVC)<br>https://qiita.com/cielavenir/items/1cfa90d73d11bb7dc3d4 (clang)|
|F#|2017/3/17|https://qiita.com/cielavenir/items/a698d6a26824ff53de81|
|Boo/Nemerle|2017/5/13|https://qiita.com/cielavenir/items/e2a783f0fe4b0fe0ed48|
|Perl6|2017/5/15|https://qiita.com/cielavenir/items/656ea17fa96c865c4498|
|Kotlin|2017/5/25|https://qiita.com/cielavenir/items/9c46ce8d9d12e51de285|
|Crystal|2018/5/8|https://qiita.com/cielavenir/items/1815bfa6a860fd1f90db|
|MoonScript|2018/6/16|https://qiita.com/cielavenir/items/8b03cce0386f4537b5ad|
|Julia/Rust|2018/12/20|https://qiita.com/cielavenir/items/3ddf72b06d625da0c4a5|
|Nim|2018/12/26|https://qiita.com/cielavenir/items/5728944867e609fd52a7|
|Tcl|2018/12/31|https://qiita.com/cielavenir/items/76cbd9c2022b48c9a2c9|
|Pascal/Cobra|2019/1/16|https://qiita.com/cielavenir/items/81b81baf8dfc1f877903|
|Icon|2019/1/17|https://qiita.com/cielavenir/items/889622dcc721f5a4da24|
|Swift|2020/5/31|https://qiita.com/cielavenir/items/3b0b84a218e35d538f7f|
|Java/Groovy/Scala|2020/5/31|https://qiita.com/cielavenir/items/7f058203a8fd03b65870|
|V|2020/10/17|https://qiita.com/cielavenir/items/df30a6c101a97a713df5|
|Zig/Zen|2020/10/17|https://qiita.com/cielavenir/items/9cced9e4a94dcd70df0f|
|Pike|2020/11/2|https://qiita.com/cielavenir/items/3a8248f41611302b34fd|
|Vala/Smalltalk|2020/11/29|https://qiita.com/cielavenir/items/085dabe593cd916af5e8|
|Objective-C|2020/11/30|https://qiita.com/cielavenir/items/a1736e38789a3dd5cc5a|
|Ruby(Ractor)|2021/1/2|https://qiita.com/cielavenir/items/f493c6d512b63cc571cc|
|Python(_xxsubinterpreters)|2021/6/29|https://qiita.com/cielavenir/items/f1f581a055db918954f1|
|Falcon/Scheme|2021/9/5|https://qiita.com/cielavenir/items/c13d12cf44f0d17f4a94|
|Clojure/Lisp|2021/9/7|https://qiita.com/cielavenir/items/7458ba076f4e5bc0f196|
|(icbrtの実装に関する)補題|2017/5/11|整数除算であってもn/(x*y)はn/x/yに等しい(ことの証明)<br>https://qiita.com/cielavenir/items/21a6711afd6be8c18c55|

Pike 8.1より、ジェネレータが導入された。continue属性をつけると、特殊な関数を返すようになり、その関数を呼ぶと処理が進むらしい。ネット上に例が一個も転がっていなかったので少し苦労した。

あと、なぜかyieldの返し値を代入しないと正しく動かない。

```pike:tyama_hena24_enum.pike
#!/usr/bin/env pike

typedef function(mixed|void, function(mixed|void...:void)|void:int) F;

int isqrt(int n){
  if(n<=0)return 0;
  if(n<4)return 1;
  int x=0;
  int y=n;
  for(;x!=y&&x+1!=y;){
    x=y;
    y=(n/y+y)/2;
  }
  return x;
}

int icbrt(int n){
  if(n<0)return -icbrt(-n);
  if(n==0)return 0;
  if(n<8)return 1;
  int x=0;
  int y=n;
  for(;x!=y&&x+1!=y;){
    x=y;
    y=(n/y/y+y*2)/3;
  }
  return x;
}

continue int generate(){
  int i=0;
  i++;
  for(;;){
    auto _=yield(i);
    i++;
  }
}

continue int drop_prev(function(int:bool) check, function(mixed|void, function(mixed|void...:void)|void:int) prev){
  int a=prev();
  int b=prev();
  for(;;){
    if(!check(b))auto _2=yield(a);
    a=b;
    b=prev();
  }
}

continue int drop_next(function(int:bool) check, F|void prev){
  int a=prev();
  int b=prev();
  auto _1=yield(a);
  for(;;){
    if(!check(a))auto _2=yield(b);
    a=b;
    b=prev();
  }
}

continue int drop_n(function(int,int:bool) check, int n, function(mixed|void, function(mixed|void...:void)|void:int) prev){
  int i=0;
  for(;;){
    i++;
    int a=prev();
    if(!check(i,n))auto _2=yield(a);
  }
}

bool is_sq(int n){return isqrt(n)**2==n;}
bool is_cb(int n){return icbrt(n)**3==n;}
bool is_multiple(int i, int n){return i%n==0;}
bool is_le(int i, int n){return i<=n;}

F f_S(F prev){return drop_next(is_sq,prev);}
F f_s(F prev){return drop_prev(is_sq,prev);}
F f_C(F prev){return drop_next(is_cb,prev);}
F f_c(F prev){return drop_prev(is_cb,prev);}
F f_h(F prev){return drop_n(is_le,100,prev);}
F f_2(F prev){return drop_n(is_multiple,2,prev);}
F f_3(F prev){return drop_n(is_multiple,3,prev);}
F f_4(F prev){return drop_n(is_multiple,4,prev);}
F f_5(F prev){return drop_n(is_multiple,5,prev);}
F f_6(F prev){return drop_n(is_multiple,6,prev);}
F f_7(F prev){return drop_n(is_multiple,7,prev);}
F f_8(F prev){return drop_n(is_multiple,8,prev);}
F f_9(F prev){return drop_n(is_multiple,9,prev);}

int main(int argc, array(string) argv){
  mapping(int:function(F:F)) m=([
    'S':f_S,
    's':f_s,
    'C':f_C,
    'c':f_c,
    'h':f_h,
    '2':f_2,
    '3':f_3,
    '4':f_4,
    '5':f_5,
    '6':f_6,
    '7':f_7,
    '8':f_8,
    '9':f_9,
  ]);
  for(;string s=Stdio.stdin.gets();){
    F z=generate();
    foreach(s; int i; int c){
      z=m[c](z);
    }
    for(int i=0;i<10;i++){
      if(i)Stdio.stdout.write(",");
      int k=z();
      Stdio.stdout.write(sprintf("%d",k));
    }
    Stdio.stdout.write("\n");
    //Stdio.stdout.flush(); // not buffered
  }
}
```